DP-1304

AEC RESEARCH AND DEVELOPMENT REPORT

EGAD - A COMPUTER PROGRAM TO COMPUTE DOSE INTEGRALS FROM EXTERNAL GAMMA EMITTERS

R. E. COOPER

DUPONT

Savannah River Laboratory

Aiken, South Carolina

Printed in the United States of America
Available from
National Technical Information Service
U. S. Department of Commerce
5285 Port Royal Road
Springfield, Virginia 22151
Price: Printed Copy $3.00; Microfiche $0.95

663341

DP-1304
Mathematics & Computers
(TID-4500, UC-32)

EGAD - A COMPUTER PROGRAM TO COMPUTE DOSE INTEGRALS FROM EXTERNAL GAMMA EMITTERS

by

R. E. Cooper

Approved by

B. C. Rusche, Director
Environmental Analysis and Planning

September 1972

E. I. DU PONT DE NEMOURS & COMPANY
SAVANNAH RIVER LABORATORY
AIKEN, S. C. 29801

CONTRACT AT(07-2)-1 WITH THE
UNITED STATES ATOMIC ENERGY COMMISSION

ABSTRACT

EGAD is an external gamma dose code programmed in FORTRAN IV that provides an estimate of whole body gamma dose resulting from radioactivity released to the environs. To model average exposure over long time periods, the distribution of material is assumed to be Gaussian about a given release height in the vertical direction with a homogeneous distribution in the crosswind direction. The code provides output in the form of dose integrals that are functions of the vertical dispersion parameter σ_z, release height, inversion lid, and gamma energy. The calculated integrals are independent of the source, radioactive decay, wind speed, and sector width. In the Savannah River man-rem evaluation program, gamma dose may be calculated at a rate of >20,000 individual cases per second on the IBM 360/65, including processing time for the meteorological data involved.

CONTENTS

	Page
Introduction	5
Statement of Problem	7
Analytical Description	7
Input Requirements	13
Output	14
Application	15
Appendices	
A. Relationships Between Secant Integrals, Bickley-Naylor Functions, and Bessel Functions	17
B. Evaluation of Gamma Buildup Factors for an Energy Range 0.01 to 5.0 Mev	19
C. EGAD FORTRAN Listing and Sample Problem.	22
References	28

LIST OF TABLES AND FIGURES

Table — Page

I Experimental Buildup Factors for Various Gamma Energies and Attenuation Lengths (a) 20

II Polynomial Coefficients for Gamma Buildup Factors in Air for Various Gamma Energies. 21

Figure

1 Geometrical Relations Between a Moving Point Source Element dS and a Ground Point R 8

2 Development of Ground and Inversion Lid Reflection Components. 11

INTRODUCTION

EGAD is an External Gamma Dose code programmed in FORTRAN IV for the IBM 360/65 that is intended to provide an estimate of whole body dose resulting from radioactive material released to the environs. The material is assumed to be bounded between the ground and an inversion lid on the vertical axis and bounded in the crosswind direction between fictitious sector boundaries or some real physical constraint such as mountains on both sides of a valley. To model average exposure over long time periods, the distribution of material is assumed to be Gaussian about a given release height in the vertical direction with a homogeneous distribution in the crosswind direction.

The code provides output in the form of dose integrals that are a function of σ_z (the vertical dispersion parameter), release height, inversion lid, and gamma energy. The calculated integrals are independent of the source, radioactive decay, wind speed, and sector width. It is therefore feasible to cover a large variety of possibilities by using a relatively small set of tabular values as a function of σ_z and employing an interpolation procedure. The code is expected to be used primarily in the construction of these tabular values to be used in a more comprehensive program of computing radiation dose. The code is currently being used in this manner as a part of the overall man-rem evaluation program at Savannah River and permits gamma dose calculations to be performed at a rate in excess of 20,000 individual cases per second on the IBM 360/65, including the processing time for the meteorological data involved.

STATEMENT OF THE PROBLEM

An assessment of the environmental effect of a nuclear installation is most acceptably accomplished by the utilization of comprehensive meteorological data covering a period from 1 to 2 years. Such a program will generally involve the processing of a very large number of jointly occurring meteorological events to estimate radiological doses from actual and hypothesized releases to the environs. A high-speed computer is necessary, and the computer code for data processing must efficiently minimize computer time for economy.

The most difficult area of radiological calculations is whole body gamma dose resulting from immersion within a cloud of material or its passage nearby. The evaluation of the associated gamma attenuation integrals as a function of material distribution in space and time accounts for most of the computer time involved. The computer code EGAD was developed to provide a means of efficiently evaluating these space integrals for man-rem calculations involving long-term averaged azimuthal distributions. The material is therefore assumed to be homogeneously distributed within sector boundaries with a gaussian distribution in the vertical direction. The material is also confined between the ground and an inversion lid at some height above the release elevation.

ANALYTICAL DESCRIPTION

EGAD is developed by utilizing the same basic assumption and mathematical approach employed by L. M. Arnett[1] to model whole body gamma dose. The cloud of radioactive gas is assumed to remain constant in size and shape during its passage with respect to a dose receptor. The space distribution of radioactive material is then described by point sources moving parallel to each other in a speed and direction dictated by the effective wind speed $\bar{u}$.

With reference to Figure 1, the attenuation A of gamma photons at a receptor R resulting from a point source element dS at K is

$$A(r) = \frac{dS \exp(-\mu r)}{4\pi r^2}, \quad r^2 = a^2 + \bar{u}^2 t^2$$

where μ = The air attenuation coefficient, m^{-1}

a = The perpendicular distance from the dose receptor R to the line described by the moving point source

$\bar{u}$ = Wind speed, m/sec

t = Time, sec

K = Moving point source

The intensity of photons (photons/m^2-sec) at the receptor is then given as

$$I(t) = \frac{B\ dS\ \exp[-\mu(a^2 + \bar{u}^2t^2)^{1/2}]}{4\pi(a^2 + \bar{u}^2t^2)}$$

where B = The buildup factor in air for gamma photons



FIGURE 1 Geometrical Relations Between a Moving Point Source Element dS and a Ground Point R

The total Dose for all t from the point source element assuming the receptor to be at a sufficient downwind distance x* ($\mu x > 10$) becomes

$$D(a) = \int_{-\infty}^{+\infty} I(t)dt$$

$$= \frac{\nu\ dS}{4\pi} \int_{-\infty}^{+\infty} B\ \frac{\exp[-\mu(a^2 + \bar{u}^2t^2)^{1/2}]\ dt}{a^2 + \bar{u}^2t^2}$$

* This restriction may be removed for long-term average estimates if azimuthal homogeneity is assumed.

ν = Dose conversion factor, (rem/sec)/(photons/m^2 - sec)

Upon substituting $(a^2 + \bar{u}^2 t^2)^{1/2}/a = \sec \phi$

$$D(a) = \frac{\nu \, dS}{2\pi \bar{u} a} \int_0^{\pi/2} B \exp[-\mu a \sec \phi] \, d\phi$$

To estimate the total contribution from a cloud, this expression must be integrated over the y-z plane in which a is imbedded. The space distribution of material must therefore be expressed as f (a) where

$$S = \int_{y\text{-}z \text{ plane}} f(a) \, da = 1 \text{ for a unit source}$$

and $dS = f(a) \, da$

The total dose integral now becomes

$$D = \frac{\nu}{2\pi \bar{u}} \frac{SG(\mu a)}{a}$$

where $G(\mu a)$ is the air attenuation kernel. This is essentially the same as the expression developed by Arnett.

The air attenuation kernal $G(\mu a)$ will take on different forms depending on the expression used for the buildup factor B. The form used in this application is

$$B = \sum_{i=0}^{3} \alpha_i (\mu a)^i$$

α_i = polynomial coefficients determined from experimental data

For this form,

$$G(\mu a) = \int_0^{\pi/2} [\exp(-\mu a \sec \phi) + \alpha_i (\mu a) \sec \phi \exp(-\mu a \sec \phi)$$

$$+ \alpha_2 (\mu a)^2 \sec^2 \phi \exp(-\mu a \sec \phi) + \alpha_3 (\mu a)^3 \sec^3 \phi$$

$$\exp(-\mu a \sec \phi)] \, d\phi$$

This expression is integrated analytically as described in Appendix A.

$$G(\mu a) = Ki_1(\mu a) + [\alpha_1(\mu a) + \alpha_3(\mu a)^3]\, K_0(\mu a) + [\alpha_2 + \alpha_3](\mu a)^2\, K_1(\mu a)$$

where Ki_1 is the Bickley-Naylor[2] function of order 1, and K_n is the Bessel function. The coefficients $\alpha_{1\to3}$ are a function of gamma energy. These coefficients are discussed in Appendix B and listed for gamma energies ranging from 0.01 to 5 Mev. Although the analytical form of $G(\mu a)$ involves functions that are difficult to evaluate, this form allows a computational time reduction of about 80% compared to a 16-point Gaussian numerical integration.

The space distribution of radioactive material as f(a) is assumed to be homogeneous in the horizontal and gaussian in the vertical direction. For this condition

$$dS = f(y,z) = \sqrt{\frac{2}{\pi}}\,\frac{1}{\sigma_z W\bar{u}}\,\exp\left[-\frac{(z - H)^2}{2\sigma_z{}^2}\right]$$

where $f(y,z) = f(a)$

and $a = \sqrt{y^2 + z^2}$

H = Release height, m

W = Sector width, m

This expression yields the space distribution without confinement in the z direction. To account for ground and inversion reflections, reference is made to Figure 2. The effect of the reflections is to produce a folding process such that the concentration at a given point is the result of summations over all folds. When modeling a Gaussion distribution, there will be an infinite number of folds at each downwind distance, but the attenuation between successive folds is usually sufficient to require only a small number of reflections to accurately describe the distribution at a point. This is shown in Figure 2 where all the $P_{L,i}$ and $P_{G,i}$ are assumed to lie on a straight line with P_o at a height z when expanded in the x direction. The points are actually coincident so the distribution at point P and height z is obtained by

$$\chi(P) = P_o + \sum_{i=1}^{\infty} \left(P_{L,i} + P_{G,i}\right)$$

where $\chi(P)$ = The space distribution at P

P_o = The unreflected contribution at P

$P_{L,i}$ = The contribution from the i^{th} lid reflection

$P_{G,i}$ = The contribution from the i^{th} ground reflection

which may be more specifically described by

$$\chi(P) = \sqrt{\frac{2}{\pi}} \frac{1}{\sigma_z} \left\{ \exp\left(-\alpha_o^2/2\sigma_z^2\right) + \sum_{i=1}^{\infty} \exp\left(-\alpha_{L,i}^2/2\sigma_z^2\right) + \sum_{i=1}^{\infty} \exp\left(-\alpha_{G,i}^2/2\sigma_z^2\right) \right\}$$

where the α's are expressed as in the following table. The parameter W has been omitted to make the expression independent of sector width assuming the space distribution in adjacent sectors to be similar. Also, unit wind velocity has been assumed.



FIGURE 2 Development of Ground and Inversion Lid Reflection Components

i	$\alpha_{L,i}$	$\alpha_{L,i} - \alpha_{L,i-1}$	$\alpha_{G,i}$	$\alpha_{G,i} - \alpha_{G,i-1}$
1.	(L-H)+(L-z)	-	H+z	-
2.	(L-H)+L+z	2z	H+L+(L-z)	2(L-z)
3.	(L-H)+2L+(L-z)	2(L-z)	H+2L+z	2z
4.	(L-H)+3L+z	2z	H+3L+(L-z)	2(L-z)
5.	(L-H)+4L+(L-z)	2(L-z)	H+4L+z	2z

From this table the following expressions are obtained that provide an efficient means of computer programming.

$$\alpha_o = z - H$$

$$\alpha_{L,i} = [(i+1) - \text{Ⓜ}]L - H + (-1)^i z$$

$$\alpha_{G,i} = [(i-1) + \text{Ⓜ}]L + H + (-1)^{i-1} z$$

where L = Inversion lid height, m

Ⓜ = (i+1) modulo 2 which can only take on the values 0 and 1.

Thus the final expression is obtained that is solved in EGAD.

$$D = \frac{1}{2\pi} \int_{z=0}^{L} f(z) \cdot \int_{y=0}^{\infty} \frac{G(\mu a)}{a} \, dydz$$

where $f(z) = \sqrt{\frac{2}{\pi}} \frac{1}{\sigma_z} \left\{ \exp - \left(\frac{(z - H)^2}{2\sigma_z^2} \right) + \right.$

$$\sum_{i=1}^{\infty} \exp \left[- \frac{1}{2} \left(\frac{[(i+1) - M]L - H + (-1)^i z}{\sigma_z} \right)^2 \right] +$$

$$\left. \sum_{i=1}^{\infty} \exp \left[- \frac{1}{2} \left(\frac{[(i-1) + M]L + H + (-1)^{i+1} z}{\sigma_z} \right)^2 \right] \right\}$$

Calculations are terminated when successive terms yield an insignificant contribution to the total integral.

For large values of σ_z ($\sigma_z > 2L$) a homogeneous distribution is assumed in the vertical direction and f (z) is a constant.

$$f(z) = 1/L$$

Integrations are performed in cartesian coordinates to take advantage of the fact that the distribution function is always constant for a given value of z. Integrations are performed numerically by the gaussian quadratures method.

INPUT REQUIREMENTS

Input to EGAD is kept as simple as practical to obtain the required data with a minimum of effort by the user. Input is described according to card type in the sequence demanded by the program. An input format specification is also given for each card type.

Card Type	(Format)	Input Description
1	(20A4)	TITLE - Up to eighty columns of alpha-numeric data to identify output
2	(I4)	NGAM - The number of gamma energies being considered in this problem 1 < NGAM <10
3	7E10.3	GAMEN(I) - The gamma energy for which the polynomial coefficients are provided, Mev
		GMU(I) - The air attenuation coefficient for this energy, m^{-1}
		AGAM(I) - The actual gamma energy for which dose integrals are to be obtained, Mev
		GAMEN should be chosen such that the energy difference between AGAM and GAMEN is minimized.
		AGNU(I) - The dose conversion factor for AGAM, (rem/sec)/(photons/m^2-sec)
		COF(J,I) - The polynomial coefficients for AGAM(I)

There must be NGAM cards of Type 3.

Card Type	(Format)	Input Description
4	6E10.3	HS - The effective release height, m
		HL - The inversion lid height, m
		YLIM - The integration limit for cross-wind or y direction measured from the plume centerline, m. If zero is input, the code will assign a value of 1 km.
		SINC - The value by which σ_z is incremented to obtain a dose integral table as $f(\sigma_z)$, m. The first calculated result will be for σ_z = SINC
		CRIT - Calculations for a given energy are terminated when $(D_i - D_{i-1})/D_i < CRIT$ where D is the dose integral.
		DMAX - The maximum number of σ_z increments permitted for a given energy. This may be overridden by the above convergence tests.

OUTPUT

Calculated dose integrals are provided for each gamma energy. The calculated dose integrals are the result of assuming a source term of 1 Ci released to the atmosphere. As previously noted, these integrals are also independent of sector width, wind speed, and radioactive decay. Printed output is provided up to the maximum permitted number of calculations. Calculations within a given energy may be terminated either by meeting the convergence criterion or when $\sigma_z > 2L$. In either case, the last calculated values are printed, and it may be assumed that this value remains constant for larger values of σ_z. All calculated values are stored within core according to gamma energy and σ_z value as a doubly subscripted array. In the case where the calculations for a particular energy are terminated before the array is full, the array is filled out by using the last calculated value for that energy. This is to facilitate the use of these data in other programs if it is desired to do so.

APPLICATION

EGAD will probably be most useful as a subprogram in support of a larger and more comprehensive dose calculational procedure. It is currently being used at Savannah River to generate dose integral tables as a function of σ_z and energy. These tables are used in the processing of measured meteorological data to estimate environmental effect of long-term radioactive releases. A value of σ_z, the vertical dispersion parameter, is obtained from processing the meteorological data, and from this an interpolated value of the resulting dose integral is obtained. Decay corrections as a function of the measured wind speed are also available from previous data analyses. When previous data analyses are not available, decay corrections may be easily generated according to the isotope being considered. This method is very useful where it is necessary to process a large set of meteorological data to generate estimated effects of long-term releases. In the Savannah River application, interpolated values of the gamma dose as a function of meteorology are obtained at a rate in excess of 20,000 per second including process time for the meteorological data.

EGAD is also expected to be useful in providing dose integral tables that may be used in manual efforts to estimate man-rem exposure. A relatively small number of computer calculations could serve as a basis for estimating effects for a variety of conditions if both gamma energy and σ_z are interpolated. Dose integrals are easily related to man-rem by calculating $D\lambda NQ/W\bar{u}$

where

D = Dose integral

λ = Attenuation factor due to radioactive decay

N = Population in sector

Q = Source term, Ci

W = Sector width, m

$\bar{u}$ = Effective wind speed, m/sec

A listing of the FORTRAN program is included with the subprogram necessary to evaluate the special functions described in Appendix C. A sample output is also provided to serve as a check calculation for possible users.

Computer time required to evaluate any particular dose integral is a function of the inversion lid height, release height, and σ_z. These parameters determine the number of terms of the infinite summation expression that will be computed. Typical calculations require about 400 msec as an average. Sufficient accuracy could probably be obtained with an 8-point gaussian quadrature integration instead of the 16-point scheme actually included in the program. With the 8-point scheme, the average time to compute dose integrals would be about 120 msec for the IBM 360/65.

APPENDIX A

Relationships Between Secant Integrals, Bickley-Naylor Functions, and Bessel Functions

The mathematical expression developed in EGAD to account for air attenuation and buildup of gamma photons is

$$(1)\quad G(\mu a) = \int_0^{\pi/2} \exp(-\mu a \sec\phi) + \alpha_1(\mu a)\sec\phi \exp(-\mu a \sec\phi) +$$

$$\alpha_2(\mu a)^2 \sec^2\phi \exp(-\mu a \sec\phi) + \alpha_3(\mu a)^3 \sec^3\phi \exp(-\mu a \sec\phi)$$

where μ = The linear air attenuation coefficient, m^{-1}

a = The distance from the source point to the point of exposure to gamma photon, m

This expression is a summation of ascending orders of Secant integrals which are shown to be contained in the general class of integrals described by Bickley-Naylor. From the fundamental definition:

$$(2)\quad Ki_n(x) = \int_0^{\infty} \frac{\exp(-x \cosh u)}{\cosh^n u}\, du$$

by letting $\cosh u = \sec\phi$

$$u = \cosh^{-1}\sec\phi$$

$$du = \sec\phi\, d\phi$$

the integrand is transformed to

$$(3)\quad Ki_n(x) = \int_0^{\pi/2} \sec^{1-n}\phi \exp(-x \sec\phi)\, d\phi$$

The attenuation kernal therefore becomes

$$(4)\quad G(\mu a) = Ki_1(\mu a) + \alpha_1(\mu a)Ki_0(\mu a) + \alpha_2(\mu a)^2 Ki_{(-1)}(\mu a) + \alpha_3(\mu a)^3\, Ki_{(-2)}(\mu a)$$

The modified Bessel functions of the second kind according to Ref. 3 may be expressed as

$$(5)\quad K_n(x) = \int_0^\infty \cosh(nu)\, \exp(-x \cosh u)\, du$$

Upon comparison with Equation 2 it is seen that

$$(6)\quad Ki_0(x) = K_0(x) \text{ and}$$

$$(7)\quad Ki_{(-1)}(x) = K_1(x)$$

To evaluate the $Ki_{(-2)}$ function the Ki_n recurrence formula involving four successive values of n is used

$$(8)\quad nKi_{(n+1)}(x) = (n-1)Ki_{(n-1)}(x) + x\left[Ki_{(n-2)}(x) - Ki_n(x)\right]$$

For n = -2,

$$(9)\quad Ki_{(-2)}(x) = \frac{Ki_{(-1)}(x)}{x} + Ki_0(x) = \frac{Ki(x)}{x} + K_0(x)$$

so that the final expression for the attenuation kernal is determined as

$$(10)\quad G(\mu a) = Ki_1(\mu a) + \left[\alpha_1(\mu a) + \alpha_3(\mu a)^3\right] K_0(\mu a) + \left[\alpha_2 + \alpha_3\right](\mu a)^2 K_1(\mu a)$$

APPENDIX B

Evaluation of Gamma Buildup Factors for an Energy Range 0.01 to 5.0 Mev

The accuracy of dose integrals as calculated in EGAD is highly dependent on an acceptable mathematical representation of gamma buildup factors as a function of energy. In this application reliance was placed on the use of experimental[4] data and polynomial regression analyses for analytical representation. The form of the analytical expression for the buildup factor B was taken to be

$$B = \sum_{i=0}^{n} \alpha_i (\mu a)^i$$

where μ = The linear air attenuation coefficient, m^{-1}

a = Distance from source to receptor, m

and α_i = The polynomial coefficients to be determined from experimental data

In the regression analyses, values of n=2 and n=3 were used. Results showed the third-degree polynomial to be markedly superior to the second degree. Although a use of the third-degree expression made it necessary to define an additional special function for the evaluation of the attenual kernal in EGAD (Appendix A), the new function was defined in terms of the same functions used for the second-degree expression. Increased accuracy was therefore obtained at little or no sacrifice in computing time as related to dose integral calculations.

Values of the experimental data from Reference 4 are included as Table I. In the regression analyses it was desired to minimize the percentage differences between analytical and experimental determinations of B. For most practical applications the most significant contributions to dose are obtained for values of μa that are small (<1). Due to the nature of the experimental data, inverse square weighting as a function of μa yielded the best results.

Calculated values of the α_i as a function of energy are presented as Table II. These values relate directly to the experimental data of Table I except for instances where interpolation was performed to obtain a uniformly structured set of coefficients. It was also assumed safe to extrapolate the data to 5 Mev since other experimental data indicate a smooth behavior in this energy region.

The last column gives the maximum percentage deviation from the experimental data as a function of μa. The maximum deviation for any energy is seen to be less than 5%.

Interpolation may be used to obtain data for energies intermediate to those tabulated, or alternatively, sufficient accuracy for most applications may be obtained by using the set of coefficients nearest the gamma energy of interest.

TABLE I

Experimental Buildup Factors[a] for Various Gamma Energies and Attenuation Lengths (a)

Gamma Energy, Mev	Experimental Buildup Factor for						
	μa=1	μa=2	μa=4	μa=7	μa=10	μa=15	μa=20
0.015	1.17	1.25	1.36	1.46	1.54	1.58	1.91
0.02	1.41	1.62	1.94	2.25	2.51	2.75	3.51
0.03	2.24	3.19	4.87	7.09	9.21	12.5	16.5
0.04	3.33	5.86	11.5	21.1	32.2	54.3	81.5
0.05	4.25	8.72	20.8	46.1	80.7	162	279
0.06	4.75	10.8	29.4	74.6	144	327	620
0.08	4.81	12.0	38.2	115	252	675	1463
0.10	4.48	11.4	38.5	124	285	811	1835
0.15	3.72	9.2	31.7	106	252	745	1730
0.20	3.27	7.73	25.6	83.9	196	564	1278
0.30	2.85	6.22	18.6	54.4	116	293	592
0.40	2.60	5.37	14.9	40.3	81	193	370
0.50	2.44	4.82	12.5	31.3	59.5	130	233
0.60	2.33	4.45	10.9	26.0	47.7	99.6	172
0.80	2.18	3.94	8.86	19.3	33.4	64.2	104
1.0	2.08	3.60	7.59	15.6	25.7	46.8	72.5
1.5	1.92	3.09	5.85	10.8	16.6	27.6	39.7
2.0	1.81	2.78	4.92	8.55	12.5	19.8	27.4

a. Obtained from Reference 4

TABLE II

Polynomial Coefficients for Experimental Buildup Factors in Air for Various Gamma Energies

Energy, Mev	Polynomial Coefficients[a]			Maximum Error, %
	α_1	α_2	α_3	
0.01	0.010390	0.001476	-0.00005806	
0.015	0.15203	-0.014892	0.00048165	+ 3.0
0.02	0.37474	-0.033582	0.0010654	- 4.8
0.03	1.2270	-0.062247	0.0020127	- 3.3
0.04	2.2543	0.087363	0.000024697	- 0.7
0.05	2.7914	0.50776	0.0020590	- 1.8
0.06	2.8286	0.95464	0.021414	- 1.5
0.07	2.7311	1.1531	0.065927	
0.08	2.5499	1.2031	0.11115	- 2.8
0.09	2.3605	1.1929	0.13581	
0.10	2.2857	1.0838	0.16117	- 3.6
0.15	1.8347	0.76133	0.16473	- 4.0
0.20	1.5119	0.66560	0.11792	- 2.9
0.30	1.1522	0.65758	0.037897	+ 0.4
0.40	1.0503	0.53489	0.016602	+ 0.7
0.50	0.98982	0.45070	0.0038726	- 0.7
0.60	0.96881	0.37066	0.00030405	- 0.7
0.70	0.95120	0.30658	-0.0018535	
0.80	0.94226	0.25805	-0.0025008	- 1.4
0.90	0.91047	0.22280	-0.0030863	
1.0	0.91686	0.18630	-0.0027652	- 1.6
1.5	0.85069	0.091974	-0.0019336	- 1.7
2.0	0.77928	0.050457	-0.0011975	- 1.5
3.0	0.66827	0.0085488	-0.00015847	
4.0	0.57420	-0.0061698	0.00021643	
5.0	0.50899	-0.014566	0.00046705	

a. $\alpha_0 = 1$ for all energies.

APPENDIX C

EGAD FORTRAN Listing and Sample Problem

```
                C     PROGRAM TO COMPUTE DOSE INTEGRALS FOR MATERIAL CONFINED WITHIN SECTOR
                C     BOUNDARIES AND WITH GROUND AND INVERSION LID REFLECTIONS
                C
ISN 0002              DIMENSION G(16),H(16),TITLE(20),GMU(10),COF(3,10),GAMEN(10)
ISN 0003              DIMENSION AGNU(10),SPN(10,200),AGAM(10)
                C
ISN 0004              DATA G/.529954E-2,.277125E-1,.618440E-1,.122298,.191062,.270992,
                     1 .359198,.452494,.547506,.640802,.729008,.808938,.877702,.932816,
                     2 .972288,.994700/, H/ .0135762,.0311268,.0475792,.0623145,
                     3 .0747980,.08457826,.0913017,.0947253,.0947253,.0913017,.0845783,
                     4 .0747980,.0623145,.0475793,.0311268,.0135762/
ISN 0005              CALL CLEAR (SPN,2000)
ISN 0006              PI=3.141593
ISN 0007              MSIZE=200
                C
                C     MSIZE IS THE ABSOLUTE MAXIMUM ALLOWED CALCULATIONS FOR A GIVEN
                C     ENERGY,CORRESPONDING TO THE ARRAY SIZE OF SPN
                C
                C
                C
                C
                C
ISN 0008              PI32=(2.*PI)**1.5
ISN 0009              NGAS=16
ISN 0010              CONG=3.7E10
ISN 0011              SQ2PI=SQRT(2.*PI)
ISN 0012              READ (5,101) TITLE
ISN 0013              WRITE(6,200) TITLE
ISN 0014              READ(5,102) NGAM
ISN 0015              DO 1 I=1,NGAM
ISN 0016            1 READ(5,103) GAMEN(I),GMU(I),AGAM(I),AGNU(I),(COF(J,I),J=1,3)
ISN 0017              READ(5,103) HS,HL,YLIM,SINC,CRIT,DMAX
ISN 0018              SGMX=2*HL
ISN 0019              IF(YLIM.EQ.0) YLIM=1.E3
ISN 0021              NSIG=DMAX+.01
ISN 0022              WRITE(6,203) HS,HL,YLIM,SINC,CRIT,NSIG
ISN 0023              WRITE(6,204)
ISN 0024              WRITE(6,205) (GAMEN(I),GMU(I),(COF(J,I),J=1,3),AGAM(I),AGNU(I),I=1
                     1    ,NGAM)
ISN 0025              WRITE(6,206)
                C
                C
ISN 0026              DO 30 NG=1,NGAM
ISN 0027              SIGZ=0.
ISN 0028              DSAV=SPN(NG,1)
ISN 0029              DO  20 M=1,NSIG
ISN 0030              SIGZ=SIGZ+SINC
ISN 0031              CON=2./(PI32*SIGZ)
ISN 0032              ZLIM=HS+3.*SIGZ
ISN 0033              ZMAX=ZLIM
ISN 0034              IF(ZLIM.GT.HL) ZLIM=HL
                C     GAUSSIAN QUADRATURE INTEGRATION ORDER 16 - TWO DIMENSIONAL
ISN 0036              SUM1=0.
ISN 0037              DO 10 I=1,NGAS
```

```
      SUM2=0.
      Z=G(I)*ZLIM
      IF(SIGZ.GT.SGMX) GO TO 6
      ZF1=EXP(-0.5*((Z-HS)/SIGZ)**2)
      ZSUM=ZF1
      DO 2 L=1,10
      MZ=L+1
      MZ=MOD(MZ,2)
      ZF=(((L+1-MZ)*HL-HS+(-1.)**L*Z)/SIGZ)**2
      IF((ZF.GT.20.).OR.(ZF.GT.ZMAX)) GO TO 3
      ZF2=EXP(-0.5*ZF)
    2 ZSUM=ZSUM+ZF2
    3 DO 4 L=1,10
      MZ=L+1
      MZ=MOD(MZ,2)
      ZF=(((L-1+MZ)*HL+HS+(-1.)**(L+1))/SIGZ)**2
      IF((ZF.GT.20.).OR.(ZF.GT.ZMAX)) GO TO 5
      ZF3=EXP(-0.5*ZF)
    4 ZSUM=ZSUM+ZF3
    5 ZSUM=ZSUM*CON
      GO TO 7
    6 ZSUM=1./(PI*HL)
    7 DO 8 J=1,NGAS
      Y=G(J)*YLIM
      A=SQRT(Z*Z+Y*Y)*GMU(NG)
      IF(A.GT.20.) GO TO 8
      CALL BESK(A,BK0,BK1,BKI1)
      A1=A
      A2=A*A
      A3=A1*A2
      COF1=COF(1,NG)
      COF2=COF(2,NG)
      COF3=COF(3,NG)
      GF=(BKI1+(COF1*A1+COF3*A3)*BK0+(COF2+COF3)*A2*BK1)*GMU(NG)/A
      SUM2=SUM2+GF*H(J)
    8 CONTINUE
      SUM1=SUM1+ZSUM*H(I)*SUM2
   10 CONTINUE
      SPN(NG,M)=SUM1*ZLIM*YLIM*CONG*AGNU(NG)
      DTEST=ABS((SPN(NG,M)-DSAV)/SPN(NG,M))
      DSAV=SPN(NG,M)
      IF((DTEST.GT.CRIT).AND.(M.LT.NSIG)) GO TO 20
      WRITE(6,201) AGAM(NG),SINC
      WRITE(6,202) (SPN(NG,MM),MM=1,M)
      IF(M.EQ.MSIZE) GO TO 30
      LM=M+1
      DO 15 MM=LM,MSIZE
   15 SPN(NG,MM)=SPN(NG,M)
      GO TO 30
   20 CONTINUE
   30 CONTINUE
  101 FORMAT(20A4)
  102 FORMAT(20I4)
  103 FORMAT(8E10.4)
  200 FORMAT('1',20X,20A4//)
  201 FORMAT('0DOSE INTEGRALS FOR',F6.3,' MEV AT SIGMA Z INTERVALS OF',
```

```
     1  F4.0,' METERS',/)
  202 FORMAT(1P10E13.3)
  203 FORMAT(20X,'RELEASE HEIGHT =',F5.0,'  METERS'/,
     1        20X,'INVERSION LID HEIGHT =',F6.1,' METERS'/,
     2        20X,'Y INTEGRATION LIMIT =',F6.0,' METERS'/,
     3        20X,'SIGMA Z IS INCREMENTED BY',F5.0,' METERS'/,
     4        20X,'CONVERGENCE CRITERION =',F6.4/,
     5        20X,'THE MAXIMUM NUMBER OF SIGMA INCREMENTS =',I4/)
  204 FORMAT(28X,'TABULAR ENERGY DATA',32X,'ACTUAL ENERGY DATA'/,
     1 16X,48('-'),15X,18('-')/,15X,'  E, MEV    AIR ATTEN       POLYNOMIA
     2L COEFFICIENTS',15X,'E, MEV     DOSE CONV'/,
     316X,18('-'),4X,29('-'),13X,7('-'),3X,7('-'))
  206 FORMAT(//)
  205 FORMAT(13X,1P5E11.3,10X,2E10.3)
      STOP
      END
```

```
      SUBROUTINE BESK(X,BK0,BK1,BKI1)
      DIMENSION T(12)                                                  BESK 420
      BK0=0.
      BK1=0.
      IF(X-1.)36,36,25                                                 BESK 540
   25 A=EXP(-X)                                                        BESK 550
      B=1./X                                                           BESK 560
      C=SQRT(B)                                                        BESK 570
      T(1)=B                                                           BESK 580
      DO 26 L=2,12                                                     BESK 590
   26 T(L)=T(L-1)*B                                                    BESK 600
C                                                                      BESK 620
C     COMPUTE K0 USING POLYNOMIAL APPROXIMATION                        BESK 630
C                                                                      BESK 640
   27 G0=A*(1.2533141-.1566642*T(1)+.08811128*T(2)-.09139095*T(3)      BESK 650
     2+.1344596*T(4)-.2299850*T(5)+.3792410*T(6)-.5247277*T(7)         BESK 660
     3+.5575368*T(8)-.4262633*T(9)+.2184518*T(10)-.06680977*T(11)      BESK 670
     4+.009189383*T(12))*C                                            BESK 680
   28 BK0=G0
C                                                                      BESK 720
C     COMPUTE K1 USING POLYNOMIAL APPROXIMATION                        BESK 730
C                                                                      BESK 740
   29 G1=A*(1.2533141+.4699927*T(1)-.1468583*T(2)+.1280427*T(3)        BESK 750
     2-.1736432*T(4)+.2847618*T(5)-.4594342*T(6)+.6283381*T(7)         BESK 760
     3-.6632295*T(8)+.5050239*T(9)-.2581304*T(10)+.07880001*T(11)      BESK 770
     4-.01082418*T(12))*C                                             BESK 780
   30 BK1=G1
      GO TO 60
C
   36 B=X/2.
      A=.5772157+ALOG(B)
      C=B*B
C                                                                      BESK 980
C     COMPUTE K0 USING SERIES EXPANSION                                BESK 990
C                                                                      BESK1000
   37 G0=-A                                                            BESK1010
      X2J=1.                                                           BESK1020
      FACT=1.                                                          BESK1030
      HJ=.0                                                            BESK1040
      DO 40 J=1,6                                                      BESK1050
      RJ=1./FLOAT(J)                                                   BESK1060
      X2J=X2J*C                                                        BESK1070
      FACT=FACT*RJ*RJ                                                  BESK1080
      HJ=HJ+RJ                                                         BESK1090
   40 G0=G0+X2J*FACT*(HJ-A)                                            BESK1100
   42 BK0=G0
C                                                                      BESK1140
C     COMPUTE K1 USING SERIES EXPANSION                                BESK1150
C                                                                      BESK1160
   43 X2J=B                                                            BESK1170
      FACT=1.                                                          BESK1180
      HJ=1.                                                            BESK1190
      G1=1./X+X2J*(.5+A-HJ)                                            BESK1200
      DO 50 J=2,8                                                      BESK1210
```

```
      RJ=1./FLOAT(J)                                                    BESK1230
      FACT=FACT*RJ*RJ                                                   BESK1240
      HJ=HJ+RJ                                                          BESK1250
   50 G1=G1+X2J*FACT*(.5+(A-HJ)*FLOAT(J))                               BESK1260
   52 BK1=G1
C
C     COMPUTE BICKLEY-NAYLOR FUNCTION KI OF ORDER 1
C
   60 IF(X.LT.0.5) GO TO 64
      IF(X.LT.2.4) GO TO 63
      IF(X.LT.5.9) GO TO 62
      IF(X.LT.9.4) GO TO 61
      A=EXP(-X)
      C=SQRT(1./X)
      BKI1=(1.253314*A/C)*(1.+(-.625+.803683/X)/X)
      GO TO 70
   61 SUM1=X*(X*(X*(-.1695446E-5)+.6393034E-4)-.8261898E-3)+.666866E-2
      SUM2=X*(X*(X*(X*(.002372806)-.03531575)+.2442256)-.7806482)+1.
      BKI1=SUM1/SUM2
      GO TO 70
   62 SUM1=X*(X*(X*(-.001137023)+.03061024)-.2895912)+.965721
      SUM2=X*(X*(X*(X*(.0395732)-.03979183)+.7326127)+.4511824)+1.
      BKI1=SUM1/SUM2
      GO TO 70
   63 SUM1=X*(X*(-.1108377)+.5157786)+1.484674
      SUM2=X*(X*(X*(X*(X*(.07879607)+.0195582)+.9487851)+1.2504993)+
     1      2.45835)+1.
      BKI1=SUM1/SUM2
      GO TO 70
   64 SUM1=X*(X*(X*(X*(X*(-.2710254E8)+.1344714E10)+.1667065E9)+.2989166
     1      E7)+7523.557)+1.570796
      SUM2=X*(X*(X*(X*(X*(X*(X*(X*(.6291512E9)+.7097694E9)+.1891334E10)+
     1      .1167903E10)+.1136180E9)+.1928012E7)+4796.828)+1.
      BKI1=SUM1/SUM2
   70 RETURN
C
      END                                                               BESK1300
```

APPENDIX C (continued)

CHECK PROBLEM ON GAMMA CALCULATIONS FOR SECTOR AVERAGED APPLICATION

RELEASE HEIGHT = 70. METERS
INVERSION LID HEIGHT = 300.0 METERS
Y INTEGRATION LIMIT = 1200. METERS
SIGMA Z IS INCREMENTED BY 50. METERS
CONVERGENCE CRITERION =0.0001
THE MAXIMUM NUMBER OF SIGMA INCREMENTS = 100

TABULAR ENERGY DATA					ACTUAL ENERGY DATA	
E, MEV	AIR ATTEN	POLYNOMIAL COEFFICIENTS			E, MEV	DOSE CONV
1.000E 00	8.200E-03	9.169E-01	1.863E-01	-2.765E-03	1.000E 00	4.960E-14
2.000E 00	5.750E-03	7.793E-01	5.046E-02	-1.197E-03	2.000E 00	8.820E-14
5.000E 00	3.550E-03	5.090E-01	-1.457E-02	4.670E-04	5.000E 00	1.610E-13

DOSE INTEGRALS FOR 1.000 MEV AT SIGMA Z INTERVALS OF 50. METERS

1.734E-03 1.501E-03 1.145E-03 9.430E-04 8.483E-04 8.116E-04 7.989E-04 7.941E-04 7.918E-04 7.905E-04
7.896E-04 7.891E-04 7.872E-04 7.872E-04

DOSE INTEGRALS FOR 2.000 MEV AT SIGMA Z INTERVALS OF 50. METERS

3.370E-03 2.963E-03 2.273E-03 1.880E-03 1.693E-03 1.620E-03 1.594E-03 1.583E-03 1.579E-03 1.576E-03
1.574E-03 1.573E-03 1.569E-03 1.569E-03

DOSE INTEGRALS FOR 5.000 MEV AT SIGMA Z INTERVALS OF 50. METERS

6.931E-03 6.219E-03 4.811E-03 4.000E-03 3.608E-03 3.450E-03 3.392E-03 3.369E-03 3.358E-03 3.351E-03
3.347E-03 3.344E-03 3.335E-03 3.335E-03

REFERENCES

1. L. M. Arnett, "Calculation of Radiation Dose From A Cloud of Radioactive Gases." *Nucl. Appl. 3*, 217 (1967).

2. W. G. Bickley and J. Naylor. "A Short Table of The Functions $Ki_n(x)$, from N=1 to N=16." *Phil. Mag. 20*, 343 (1935).

3. *Engineering Compendium on Radiation Shielding*. R. G. Jaeger, Ed., Vol. 1, p. 218 (Berger's Data), Springer-Verlag, New York (1968).

4. M. Abramowitz and I. A. Stegum. *Handbook of Mathematical Functions*. Applied Mathematical Series 55, p. 376, National Bureau of Standards, U. S. Geological Printing Office, Washington, D. C. (1964).